

Kapitel V

Grafiken mit Beschriftungen

In diesem Kapitel:

| | |
|--|---|
| Wissenschaftliche Figuren beschriften | 2 |
| Beschriftung und Grafik im Positionsrahmen | 6 |
| Zusammenfassung | 9 |

Im Kapitel 5 wurden die Grundlagen der Arbeit mit dem Word-Objektmodell vorgestellt. Diese Lösung veranschaulicht die folgende Objekte.

Im Blickpunkt:**Word-Objekte**

Dialog
Field
Frame
InlineShape
PageSetup
Range
Selection
Table

Seit der Version 97 fügt Word Beschriftungen für Shape-Objekte (das Word-Objektmodell ist eingehend in Kapitel 5 erläutert) standardmäßig in Textfelder ein. Daraus entstehen zwei grundsätzliche Nachteile:

- Das Textfeld ist nicht mit dem grafischen Objekt verbunden. Wird Letzteres im Dokument verschoben, wandert das Textfeld nicht automatisch mit. Dieses Verhalten verwirrt den Benutzer und kostet viel Zeit.
- Beschriftungen in Textfeldern werden bei der automatischen Erstellung von Verzeichnissen und Querverweisen von Word nicht erkannt. Folglich fehlen diese Einträge in den Verzeichnissen. Es hilft also nicht, Grafik und Beschriftung in einem Textfeld oder Zeichnungsbereich zu vereinen.

Für das erste Problem gibt es mehrere Lösungswege. Für das zweite aber ist unabdingbar, dass die Beschriftung Teil des Dokumenttextes ist. Folglich muss die Grafik ein `InlineShape` sein, damit sie mit der Beschriftung zusammengehalten wird. Womit wir ein drittes Problem haben: Wie kann die Textflussformatierung trotzdem beibehalten werden?

Hier bieten sich zwei Alternativen an: die Grafik samt Beschriftung entweder in eine Tabelle oder in einen Positionsrahmen einzufügen. Beide Lösungen stellen wir in diesem Kapitel vor, wobei jene für die Tabelle eine Beschriftung auf wissenschaftliche Art, rechts neben der Grafik, erstellt.

Wissenschaftliche Figuren beschriften

Die Benutzerschnittstelle von Word fügt Beschriftungen entweder ober- oder unterhalb des grafischen Objekts ein. Zwar können sie nachträglich verschoben werden, aber der Benutzer soll sich möglichst dem Dokumentinhalt widmen und nicht durch solche Nebensächlichkeiten abgelenkt werden. Deshalb ist es sinnvoll, ihm in der Vorlage für wissenschaftliche Arbeiten ein Werkzeug bereitzustellen, das ihm diese gestalterische Arbeit abnimmt.


Die Lösung in Listing V.1 fordert zunächst den Benutzer auf, eine Beschriftung einzugeben, fügt diese dann in die rechte Spalte einer Tabelle ein und startet schließlich den Formel-Editor, wie in Abbildung V.1 dargestellt.

HINWEIS

Es ist nicht möglich, mit Office-Bordmitteln programmtechnisch eine Formel im Formel-Editor zu erstellen, denn der Formel-Editor bietet keine Automatisierungsschnittstelle an. Sie müssten sich bei entsprechendem Bedarf nach einer Anwendung von Drittherstellern umsehen. Die Firma MathType hat den Formel-Editor an Microsoft lizenziert und wäre eine mögliche Kontaktstelle (<http://www.mathtype.com>).

Abbildg. V.1

In wissenschaftlichen Arbeiten muss die Beschriftung oft rechts neben der Figur stehen.



Die Hauptprozedur *FormelEinfuegenUndBeschriften* weist die gegenwärtige Markierung einem Range-Objekt zu und kontrolliert, ob sie sich an einer für die Tabelle geeigneten Stelle befindet. Dies darf weder in einer bestehenden Tabelle noch mitten in einem Absatz sein. In beiden Fällen wird die Markierung zum Ende der Tabelle oder des Absatzes bewegt und ein neuer Absatz eingefügt. Erst dann wird die neue Tabelle erstellt.

Dies geschieht mit Hilfe der Funktion *TabelleErstellen*, die als Argument das Range-Objekt des Zielortes für die Tabelle erwartet. Eine Tabelle mit zwei Spalten und einer Reihe wird eingefügt. Befindet sich diese Tabelle unmittelbar unter einer bestehenden, betrachtet Word die beiden als Einheit, wodurch die Tabelle plötzlich mehr als eine Zeile hat. In diesem Fall wird sie von der anderen abgetrennt:

```
If tbl.Rows.Count > 1 Then tbl.Split BeforeRow:=rng.Rows(1)
```

Sie haben die Wahl, die Tabellenspalten dem Inhalt anzupassen oder umgekehrt den Inhalt dazu zu zwingen, eine vorgegebene Breite zu respektieren. Das vorliegende Beispiel nutzt die zweite Möglichkeit: Die Spaltenbreite wird beim Erstellen der Tabelle fixiert (*AutoFitBehavior:=wdAutoFitFixed*) und anschließend das Maß festgelegt (*tbl.Columns(1).Width*). Die linke Spalte für das grafische Objekt erstreckt sich über 60% der Textbreite; die rechte über die übrigen 40 %.

Zudem wird dafür gesorgt, dass die Beschriftung am unteren Rand der Tabellenzelle ausgerichtet wird (*tbl.Cell(1, 2).VerticalAlignment = wdCellAlignVerticalBottom*) und dass der Tabelleninhalt links und rechts mit dem Dokumenttext bündig steht. Hierfür wird der Abstand zwischen dem Tabellenrand und dem Inhalt entfernt (*tbl.LeftPadding = 0#*).

Die Ausführung kehrt an die rufende Hauptprozedur zurück, die als Zielbereich die Zelle für die Beschriftung festlegt und dann die Prozedur *BeschriftungEinfuegen* aufruft, die ihrerseits dieses Range-Objekt als Argument erwartet. Der Benutzer erhält nun in Form einer InputBox die Aufforderung zur Eingabe des Beschriftungstextes. Ist diese erfolgt, wird der Zielbereich mit der Formatvorlage *Beschriftung* formatiert (*rng.Style = wdStyleCaption*) sowie rechts ausgerichtet. (Die Ausrich-

tung muss nach der Zuweisung der Formatvorlage erfolgen, da diese eine bestehende Formatierung ersetzt.)

Die Beschriftung wird in drei Teilen eingefügt:

- die Bezeichnung (»Formel«)
- eine SEQ-Feldfunktion für die Nummerierung mit der Bezeichnung »Formel« (Diese Bezeichnung wird von Word benötigt, um Verzeichnisse und Querverweise zu erstellen.)
- ein Doppelpunkt als Trennzeichen, gefolgt von der Benutzereingabe, deren Fett-Formatierung anschließend entfernt wird

Interessant ist in diesem Teil der Lösung der Umgang mit dem Range-Objekt. Sie können deutlich sehen, wie es manipuliert wird, so dass jeder Teil dem vorherigen folgt (besonders nach Einfügen der Feldfunktion). Auch ist ersichtlich, wie Text gezielt formatiert wird.

Zum Schluss werden alle SEQ-Feldfunktionen im Dokument aktualisiert, damit die Nummerierung stimmt.

Zurück in der Hauptprozedur, wechselt der Zielbereich zur ersten Spalte, und die Einfügemarke wird hierher versetzt:

```
Set rng = tbl.Cell(1, 1).Range
rng.Collapse Direction:=wdCollapseStart
rng.Select
```

Zuletzt wird ein Formel-Objekt erstellt und die Kontrolle wieder dem Benutzer übergeben:

```
Set ils = rng.InlineShapes.AddOLEObject(ClassType:="Equation.3", FileName:="",
LinkToFile:=False, DisplayAsIcon:=False)
```

Listing V.1

Eine Beschriftung rechts ausgerichtet in einer Tabelle einfügen; das Objekt steht links daneben

```
Option Explicit

Sub FormelEinfuegenUndBeschriften()
    Dim rng As Word.Range
    Dim ils As Word.InlineShape
    Dim tbl As Word.Table
    Dim strBeschriftungsText As String

    strBeschriftungsText = InputBox("Bitte die Beschriftung eingeben")
    'Prüfen, ob keine Beschriftung eingegeben oder Abbrechen betätigt wurde
    If Len(strBeschriftungsText) = 0 Then Exit Sub

    Set rng = Selection.Paragraphs(1).Range
    'Falls die Markierung in einer Tabelle steht
    'sie ausserhalb der Tabelle verschieben
    If rng.Information(wdWithInTable) Then
        Set rng = rng.Tables(1).Range
        rng.Collapse Direction:=wdCollapseEnd
        rng.InsertAfter vbCr
        rng.Collapse Direction:=wdCollapseEnd
    End If
```

Listing V.1 Eine Beschriftung rechts ausgerichtet in einer Tabelle einfügen; das Objekt steht links daneben (*Fortsetzung*)

```
'Falls die Markierung in einem Absatz mit Text steht,
'die Tabelle in einen leeren einfügen, der dem
'aktuellen folgt
'Testen, ob der aktuellen Absatz "leer" ist
If Len(rng.Paragraphs(1).Range.Text) > 1 Then
    rng.Collapse Direction:=wdCollapseEnd
    rng.InsertBefore vbCr
    rng.Collapse Direction:=wdCollapseStart
End If

Set tbl = TabelleErstellen(rng)
'Zuerst die Beschriftung rechts einfügen, da der Benutzer
'im Abschluss die Formel eingeben muss
Set rng = tbl.Cell(1, 2).Range
rng.Collapse Direction:=wdCollapseStart
BeschriftungEinfuegen rng, strBeschriftungsText

'Der Formel-Editor in der ersten Spalte starten
Set rng = tbl.Cell(1, 1).Range
rng.Collapse Direction:=wdCollapseStart
rng.Select
Set ils = rng.InlineShapes.AddOLEObject(ClassType:="Equation.3", FileName="", _
    LinkToFile:=False, DisplayAsIcon:=False)
End Sub

Private Function TabelleErstellen(ByRef rng As Word.Range) As Word.Table
    Dim tbl As Word.Table
    Dim pgs As Word.PageSetup
    Dim sngTextBreite As Single
    Dim cel As Word.Cell

    'Die Tabelle einfügen. Spalte links für die Formel,
    'Spalte rechts für die Beschriftung.
    'Die Formel wird sich der Breite der linken Spalte anpassen
    Set tbl = rng.Tables.Add(Range:=rng, NumRows:=1, _
        NumColumns:=2, AutoFitBehavior:=wdAutoFitFixed)
    'Sicherstellen, dass sich die neue Tabelle nicht mit einer anderen
    'direkt daneben befindlichen verbunden hat
    If tbl.Rows.Count > 1 Then tbl.Split BeforeRow:=rng.Rows(1)
    'Die erste Spalte soll 2/3 der Seite, zwischen den Seitenränder, breit sein
    Set pgs = rng.Sections(1).PageSetup
    sngTextBreite = (pgs.PageWidth - pgs.LeftMargin - pgs.RightMargin)
    tbl.Columns(1).Width = 0.6 * sngTextBreite
    tbl.Columns(2).Width = 0.4 * sngTextBreite
    'Beschriftung unten in der Tabellenzelle ausrichten
    tbl.Cell(1, 2).VerticalAlignment = wdCellAlignVerticalBottom
    'Der Text soll bündig mit dem Dokumenttext stehen
    tbl.LeftPadding = 0#
    tbl.RightPadding = 0#
    Set TabelleErstellen = tbl
End Function

Private Sub BeschriftungEinfuegen(ByRef rng As Word.Range, strBeschriftungsText)
    Dim fld As Word.Field
```

Listing V.1 Eine Beschriftung rechts ausgerichtet in einer Tabelle einfügen; das Objekt steht links daneben (*Fortsetzung*)

```
'Die Zelle mit der Beschriftung-Formatvorlage formatieren
rng.Style = wdStyleCaption
'Beschriftung rechts ausrichten
rng.Paragraphs.Alignment = wdAlignParagraphRight
rng.Text = "Formel "
rng.Collapse Direction:=wdCollapseEnd
'Nach Bezeichnung eine SEQ-Feldfunktion für die Nummerierung einfügen
Set fld = rng.Fields.Add(Range:=rng, Text:="SEQ Formel", _
PreserveFormatting:=False)
'Nach der Feldfunktion kommt die Beschriftung, die nicht mehr fett ist
Set rng = fld.Result
rng.Collapse Direction:=wdCollapseEnd
rng.MoveStart Unit:=wdCharacter, Count:=1
rng.Text = ": " & strBeschriftungsText
rng.Font.Bold = False
'Alle SEQ-Feldfunktionen aktualisieren
For Each fld In rng.Parent.Fields
    If fld.Type = wdFieldSequence Then fld.Update
Next
End Sub
```

Beschriftung und Grafik im Positionsrahmen

Für das Beispiel in Abbildung V.2 wurde eine Grafik samt zugehöriger Beschriftung in einen Positionsrahmen eingefügt. Wird dieser verschoben, wandern Grafik und Beschriftung mit. Die Prozedur in Listing V.2 passt dabei die Größe des Positionsrahmens der Grafik an. Es ist aber umgekehrt auch möglich, die Größe der Grafik dem Behälter anzupassen: Erstellen Sie zuerst den Positionsrahmen und sorgen Sie dafür, dass sich der Zielbereich im Positionsrahmen befindet. Fügen Sie dann die Grafik als `InLineShape` ein.

Abbildg. V.2 Die rechtsbündige Grafik steht zusammen mit ihrer Beschriftung in einem Positionsrahmen.

Mit der Prozedur *GrafikInPosRahmenMitBeschriftung* (die Symbolschaltfläche "mit Textfluss") wird sie zusammen mit der Grafik in einen Positionsrahmen, rechts ausgerichtet mit Textflussformatierung, eingefügt. Grafik und Beschriftung bleiben damit zusammen, wenn der Positionsrahmen verschoben wird. Zudem erscheint die Beschriftung im Inhaltsverzeichnis.



Zuerst prüft die Prozedur *GrafikInPosRahmenMitBeschriftung*, ob der Zielbereich (Absatz, worin sich die Einfügemarke befindet) das Einfügen eines Positionsrahmens unterstützt. Das bedeutet, der Absatz muss sich im Dokumenttextkörper befinden und darf nicht in einer Tabelle, Endnote, Fußnote oder Ähnlichem enthalten sein. (Beachten Sie hier die Verwendung der Eigenschaft *Selection.Type*.) Sofern kein Problem gefunden wird, läuft die Ausführung weiter, ansonsten wird zum Ausstiegspunkt gesprungen.

Ist alles soweit in Ordnung, wird der Benutzer aufgefordert, eine Grafikdatei auszuwählen. Die Prozedur zeigt das Word-eigene Dialogfeld *Grafik einfügen* an mit der Methode *Display*. Wie in Kapitel 15 erläutert, wird mit dieser Methode das Dialogfeld nur angezeigt, aber nicht von Word ausgewertet. Stattdessen werden die Einstellungen für die weitere Bearbeitung in Variablen gespeichert.

Wählt der Benutzer *Abbrechen*, springt der Code zum Ausstiegspunkt. Andernfalls wird die Variable *lAuswahl* dahingehend ausgewertet, ob der Benutzer die Grafik mit oder ohne Verknüpfung einfügen möchte, und wenn mit, ob die Grafikdaten im Dokument gespeichert werden sollen. Die Werte der Variablen *bLinkToFile* und *bSaveWithDoc* werden entsprechend festgelegt und in der *AddPicture*-Methode verwendet.

Die eingefügte Grafik wird markiert, dann der Positionsrahmen eingefügt und rechts ausgerichtet (*fram.HorizontalPosition = wdFrameRight*). Es fällt auf, dass sich diese Prozedur ausschließlich des *Selection*- und nicht des *Range*-Objekts bedient. Das hat zwei Gründe:

- Wir brauchen zu Beginn der Prozedur die Eigenschaft *Selection.Type*, um festzustellen, ob der Zielbereich gültig ist.
- Der Positionsrahmen orientiert sich an der gegenwärtigen Markierung, dies obwohl die Methode *Frames.Add* ein *Range*-Argument hat. Soll – wie in diesem Fall – bestehender Text in einen Positionsrahmen aufgenommen werden, muss mit dem *Selection*-Objekt gearbeitet werden.

Nach Einfügen des Positionsrahmens ist die Grafik immer noch markiert. Eine Eigenart des Positionsrahmens ist, dass falls er nur eine Grafik enthält, die beiden »verschmelzen«; Word »sieht« nur die Grafik. Deshalb müssen weitere Zeichen in den Positionsrahmen eingefügt werden, bevor mit dem Inhalt weitergearbeitet werden kann. Da sich die Beschriftung unterhalb der Grafik befinden soll, wird ein Absatz nach der Grafik benötigt. Zu diesem Zweck wird die Markierung auf einen Punkt verkleinert (*Selection.Collapse wdCollapseEnd*) und der Absatz eingefügt.

Zum Schluss blendet die Prozedur das Dialogfeld *Beschriftung* ein.

Listing V.2 Eine »mit Text in Zeile« formatierte Grafik in einen Positionsrahmen einfügen

```
Sub GrafikInPosRahmenMitBeschriftung()
    Dim strBildName As String
    Dim lAuswahl As Long
    Dim bLinkToFile As Boolean
    Dim bSaveWithDoc As Boolean
    Dim ils As Word.InlineShape
    Dim fram As Word.Frame

    ' Die Markierung darf/soll nicht in einer Tabelle, Positionsrahmen o.ä. stehen.
    If (Selection.Type <> wdSelectionIP And Selection.Type <> wdSelectionNormal) _
        Or Selection.Information(wdWithInTable) Or _
        Selection.Information(wdInEndnote) Or _
        Selection.Information(wdInFootnote) Then
        MsgBox "Die Markierung muss im Dokumenttext stehen.",_
```

Listing V.2 Eine »mit Text in Zeile« formatierte Grafik in einen Positionsrahmen einfügen (Fortsetzung)

```

        vbOKOnly + vbCritical, strMSG_TITEL
        GoTo Fertigstellen
    End If
    ' Bild Dateiname, und Art der Einfügung
    With Dialogs(wdDialogInsertPicture)
        ' Wenn nicht "Einfügen" gewählt wurde, abbrechen.
        If .Display <> -1 Then GoTo Fertigstellen
        lAuswahl = .LinkToFile
        strBildName = .Name
    End With

    ' Feststellen, ob Bild verknüpft und/oder im Dokument zu speichern ist.
    Select Case lAuswahl
        Case 0
            bLinkToFile = False
            bSaveWithDoc = True
        Case 1
            bLinkToFile = True
            bSaveWithDoc = True
        Case 2
            bLinkToFile = True
            bSaveWithDoc = False
        Case Else
    End Select
    ' Bild einfügen...
    Set ils = ActiveDocument.InlineShapes.AddPicture(FileName:=strBildName, _
        LinkToFile:=bLinkToFile, SaveWithDocument:=bSaveWithDoc, Range:=Selection.Range)
    ' ...und markieren.
    ils.Select
    ' Positionsrahmen um die Markierung einfügen...
    Set fram = Selection.Frames.Add(Range:=Selection.Range)
    ' ...und rechts ausrichten.
    fram.HorizontalPosition = wdFrameRight
    ' Neuen Absatz für die Beschriftung nach dem Bild einfügen.
    Selection.Collapse wdCollapseEnd
    Selection.Text = vbCr
    ' Benutzer für die Beschriftung auffordern.
    Dialogs(wdDialogInsertCaption).Show

Fertigstellen:

End Sub

```



Die Beispieldatei *BspV_01.doc* finden Sie auf der CD-ROM zum Buch im Ordner *\Beispiele\KapV*.

Zusammenfassung

In diesem Kapitel wurden zwei Ergänzungen zu der von Word angebotener Funktionalität für Beschriftungen vorgestellt.

- Im Abschnitt »Wissenschaftliche Figuren beschriften« (Seite 2) wird erläutert, wie eine rechts ausgerichtete Beschriftung neben dem Objekt positioniert wird.
- Wie eine Beschriftung mit seinem Objekt zusammengehalten werden kann, steht im Abschnitt »Beschriftung und Grafik im Positionsrahmen« (Seite 6) beschrieben. Obwohl Beschriftung und Objekt mit Textfluss formatiert sind, wird die Beschriftung von der Verzeichnis-Funktionalität erkannt.

