

## Kapitel VIII

# Attraktive Tischkarten im Nu

### In diesem Kapitel:

Das Seriendruck-Hauptdokument	2
Die WordArt-Objekt-Vorlage	3
Die Tischkarten erstellen	5
Zusammenfassung	9

In Kapitel 5 wurden die Grundlagen der Arbeit mit dem Word-Objektmodell vorgestellt. Diese Lösung veranschaulicht die folgenden Objekte, Eigenschaften und Methoden, mit Schwergewicht auf der Arbeit mit grafischen Objekten und WordArt.

---

**Im Blickpunkt:**

Seriendruck

WordArt

Word-Objekte

Shapes

Shapes.TextEffect

MailMerge

Table

---

Tischkarten werden oft für große Anlässe gebraucht. Heutzutage werden Adresslisten mühelos in einer Word-Tabelle, in einem Excel-Arbeitsblatt oder in einer Access-Datenbank angelegt. Mit der Seriendruck-Funktionalität von Word werden persönliche Einladungen und adressierte Umschläge schnell erstellt. Warum nicht auf ähnliche Art und Weise auch Tischkarten erstellen?

Für den Word-Anwender gibt es zwei Hindernisse:

- Gewöhnlicher Text kann nur um 90 oder 270 Grad gedreht werden, nicht jedoch um 180 Grad. WordArt-Objekte lassen sich zwar beliebig drehen, aber ...
- ... es ist nicht möglich, Seriendruckfelder in WordArt-Objekte zu integrieren.

Mit einer Automatisierungslösung können Seriendruck und WordArt vereint werden, um ansprechende Tischkarten schnell und problemlos zu reproduzieren.

Aus Sicht des Anwenders erfolgt die Herstellung in drei Schritten:

- Das Seriendruck-Hauptdokument mit einer Tabelle für die Tischkarten erstellen.
- Die gewünschte WordArt-Form festlegen.
- Ein Makro ausführen, das den Seriendruck mit den Daten zusammenführt und anschließend die Daten in WordArt-Objekte übernimmt.

## Das Seriendruck-Hauptdokument

Tischkarten werden ähnlich wie Etiketten gehandhabt und in einer Word-Tabelle angeordnet. Liegt keine vorgestanzte Papierform vor, kann die Größe der Tabellenzellen selbst definiert werden – die Gitternetzlinien zeigen die Begrenzungen an. Im Falle von Tischkarten werden zwei Tabellenzellen pro Karte gebraucht; der dazwischen liegende Zellenrand dient als Falzlinie. Falls Rahmenlinien auf den Karten unerwünscht sind, können, wie die Abbildung VIII.1 zeigt, feine Striche an den linken und rechten Tabellenrändern als Falzlinien dienen.

Seriendruckfelder für den Karteninhalt werden wie üblich eingefügt. Falls ein Blatt Papier mehrere Karten enthält, wird eine *Next*-Feldfunktion am Anfang der ersten Zelle jeder neuen Karte gebraucht. Damit wird der Seriendruck gezwungen, den nächsten Datensatz auszugeben.

Abbildg. VIII.1 Ein Seriendruck-Hauptdokument für Tischkarten

«Vorname»«Nachname»¶ ...¶ «Position»»
«Vorname»«Nachname»¶ ...¶ «Position»»
«Nächster Datensatz»«Vorname»«Nachname»¶ ...¶ «Position»»
«Vorname»«Nachname»¶ ...¶ «Position»»

## Die WordArt-Objekt-Vorlage

Der Anwender soll selbst entscheiden, wie die Information auf der Tischkarte auszusehen hat, indem er ein WordArt-Objekt entwirft und als Vorlage bestimmt.

Weil das Objekt in die Tabellenzellen passen muss, soll er die Tabelle in ein neues Dokument kopieren, alle Zellen außer einer löschen und darin das WordArt-Objekt erstellen, wie in Abbildung VIII.2 dargestellt.

Dieses Dokument muss in dem gleichen Ordner wie das Seriendruck-Hauptdokument gespeichert werden, unter dem Namen »WordArt.doc«.

**HINWEIS** Die Anweisungen entsprechen dem hier verwendeten Code. Sie dürfen die Lösung selbstverständlich den eigenen Bedürfnissen anpassen.

Abbildg. VIII.2 Ein WordArt-Objekt in einer einzelnen Tabellenzelle dient als Vorlage für die Tischkarten.




Bei der Positionierung des Objekts muss darauf geachtet werden, dass es

- mit Textfluss formatiert wird (also *nicht* »mit Text in Zeile«),
- senkrecht relativ zur Tabellenzelle und nicht zur Seite positioniert wird
- und verankert wird.

Die Abbildung VIII.3 veranschaulicht die Positionierungseinstellungen.

**HINWEIS** Mehr über die Arbeit mit Grafiken (Shapes) steht im entsprechenden Abschnitt von Kapitel 5 beschrieben.

Abbildg. VIII.3 Die Positionierungseinstellungen für das WordArt-Objekt



# Die Tischkarten erstellen

Jetzt ist nur noch das Makro aus Listing VIII.1 auszuführen, um die Tischkarten zu erstellen (Abbildung VIII.4). Wir schlagen vor, es einer Symbolschaltfläche in einer Vorlage zuzuweisen, die als Add-In geladen wird.

Die Prozedur *TischkartenErstellen* kontrolliert zuerst, ob das aktuelle Dokument ein Seriendruck-Hauptdokument ist; falls nicht, wird abgebrochen.

Andernfalls wird das Dokument gespeichert, der Dokumentpfad festgehalten (um später *WordArt.doc* öffnen zu können) und eine Dokumentvariable mit dem aktuellen Zeitpunkt angelegt. Erst dann wird der Seriendruck in ein neues Dokument ausgeführt.

Da *MailMerge.Execute* das Ergebnisdokument des Seriendrucks nicht zurückgibt, erfolgt eine Prüfung der Dokumentvariablen, die durch das Ergebnisdokument geerbt wird. Ist sie nicht vorhanden oder hat sie einen abweichenden Wert, wird der Vorgang abgebrochen. Ansonsten wird das Seriendruck-Hauptdokument geschlossen und das Erstellen der WordArt-Objekte im Ergebnisdokument fortgesetzt.

*WordArt.doc* wird unsichtbar geöffnet und das einzige grafische Objekt erfasst. Danach wird durch jede Tabellenzelle des Ergebnisdokuments geschleift. (Das Resultatdokument kann mehrere Tabellen enthalten, wenn der Seriendrucktyp nicht »Verzeichnis« ist.) Der Zelleninhalt wird in einer Variablen festgehalten, dann aus der Zelle gelöscht. Das WordArt-Objekt wird durch die Prozedur *WordArtErstellen* eingefügt.

Ihr werden die WordArt-Vorlage, der Textinhalt der Zelle, der Zellenbereich sowie die Angabe, ob das WordArt-Objekt gedreht werden soll, übergeben. Die Prozedur prüft jede Shape- und TextEffect-Eigenschaft der WordArt-Vorlage und weist sie dem neu erstellten Objekt zu.

Zum Schluss wird das WordArt-Vorlage-Dokument geschlossen und das Ergebnisdokument aktiviert.

**ACHTUNG** Die Ausführung kann wegen der vielen grafischen Formatierungen recht lange dauern.

**Listing VIII.1** Den Seriendruck zusammenführen und im Ergebnisdokument WordArt-Objekte einfügen

```
Sub TischkartenErstellen()
    Dim strPfad As String
    Dim docSeriendruck As Word.Document
    Dim docResultat As Word.Document
    Dim docWordArt As Word.Document
    Dim strDokVar As String
    Dim strDokVarInhalt As String

    If ActiveDocument.MailMerge.MainDocumentType = _
        wdNotAMergeDocument Then
        MsgBox "Das aktuelle Dokument muss ein Seriendruckdokument " & _
            "mit verbundener Datenquelle sein.", vbCritical + vbOKOnly
        Exit Sub
    End If

    Set docSeriendruck = ActiveDocument
    'Das Seriendruck-Hauptdokument vor der Zusammenführung speichern.
```

**Listing VIII.1** Den Seriendruck zusammenführen und im Ergebnisdokument WordArt-Objekte einfügen *(Fortsetzung)*

```

docSeriendruck.Save
strPfad = docSeriendruck.Path
strDokVar = "WordArt"
'Die genaue Zeit als Dokumentvariable speichern, so dass nachgeprüft
'werden kann, ob am Schluss das Ergebnisdokument das aktuelle ist.
strDokVarInhalt = Format(Date, "yyyymmdd") & Format(Time, "hhmmss")
docSeriendruck.Variables(strDokVar) = strDokVarInhalt

docSeriendruck.MailMerge.Destination = wdSendToNewDocument
docSeriendruck.MailMerge.Execute
Set docResultat = ActiveDocument
'Sicherstellen, dass das aktuelle Dokument das Seriendruckergebnis ist.
'Es wird die sich im Hauptdokument befindenden Variablen 'erben'.
If docResultat.Variables(strDokVar) <> strDokVarInhalt Then
    MsgBox "Das aktuelle Dokument ist nicht das Seriendruckergebnis." &
        "Der Vorgang wird abgebrochen. Die Tischkarten konnten nicht erstellt werden.", _
        vbCritical + vbOKOnly
    Exit Sub
End If
docSeriendruck.Close SaveChanges:=wdDoNotSaveChanges

Dim shp As Word.Shape
Set docWordArt = Application.Documents.Open(Filename:=strPfad & "\WordArt.doc", _
    Visible:=False)
Set shp = docWordArt.Shapes(1)

Dim tbl As Word.Table
Dim cel As Word.Cell
Dim rng As Word.Range
Dim bDrehen As Boolean
Dim strText As String

For Each tbl In docResultat.Tables
    For Each cel In tbl.Range.Cells
        Set rng = cel.Range
        If cel.RowIndex Mod 2 = 0 Then
            bDrehen = False
        Else
            bDrehen = True
        End If
        strText = TrimZelle(rng.Text)
        rng.Delete
        WordArtErstellen shp, strText, rng, bDrehen
    Next
Next
docWordArt.Close SaveChange:=wdDoNotSaveChanges
docResultat.Activate
End Sub

```

Listing VIII.1 Den Seriendruck zusammenführen und im Ergebnisdokument WordArt-Objekte einfügen (Fortsetzung)

```

Sub WordArtErstellen(shp As Word.Shape, strText As String, _
    rng As Word.Range, bDrehen As Boolean)

    Dim shpNeu As Word.Shape
    Dim te As Word.TextEffectFormat
    Dim lPresetEffect As Long

    Set te = shp.TextEffect
    '-2 bei Shapes bedeutet allgemein, dass keine Auswahl getroffen wurde.
    'Dieser Wert kann von einem anderen Objekt nicht übernommen werden.
    'Deshalb treffen wir eine Wahl.
    If te.PresetTextEffect = -2 Then
        lPresetEffect = 1
    Else
        lPresetEffect = te.PresetTextEffect
    End If
    'Das neue WordArt-Objekt mit den Eigenschaften der Vorlage formatieren.
    Set shpNeu = rng.Document.Shapes.AddTextEffect( _
        PresetTextEffect:=lPresetEffect, Text:=strText, _
        FontName:=te.FontName, FontSize:=te.FontSize, FontBold:=te.FontBold, _
        FontItalic:=te.FontItalic, Left:=shp.Left, Top:=shp.Top, Anchor:=rng) _
    shpNeu.Height = shp.Height
    shpNeu.Width = shp.Width
    With shpNeu.TextEffect
        .PresetShape = te.PresetShape
        .KernedPairs = te.KernedPairs
        .NormalizedHeight = te.NormalizedHeight
        .RotatedChars = te.RotatedChars
        .Tracking = te.Tracking
    End With
    With shpNeu.Fill
        .BackColor = shp.Fill.BackColor
        .ForeColor = shp.Fill.ForeColor
        Select Case shp.Fill.GradientColorType
            Case msoGradientOneColor
                .OneColorGradient shp.Fill.GradientStyle, _
                    Variant:=shp.Fill.GradientVariant, _
                    Degree:=shp.Fill.GradientDegree
            Case msoGradientTwoColors
                .TwoColorGradient Style:=shp.Fill.GradientStyle, _
                    Variant:=shp.Fill.GradientVariant
            Case msoGradientPresetColors
                .PresetGradient Style:=shp.Fill.GradientStyle, _
                    Variant:=shp.Fill.GradientVariant, _
                    PresetGradientType:=shp.Fill.PresetGradientType
            Case msoGradientColorMixed
        End Select
        If shp.Fill.Pattern <> -2 Then
            .Patterned shp.Fill.Pattern
        End If
        If shp.Fill.TextureType = msoTexturePreset And _
            shp.Fill.PresetTexture <> -2 Then
            .PresetTextured shp.Fill.PresetTexture
        End If
        .Transparency = shp.Fill.Transparency
    End With
End With

```

**Listing VIII.1** Den Seriendruck zusammenführen und im Ergebnisdokument WordArt-Objekte einfügen (*Fortsetzung*)

```

With shpNeu.Line
    .BackColor = shp.Line.BackColor
    .DashStyle = shp.Line.DashStyle
    .ForeColor = shp.Line.ForeColor
    .Style = shp.Line.Style
    .Transparency = shp.Line.Transparency
    .Weight = shp.Line.Weight
End With
With shpNeu.Shadow
    .ForeColor = shp.Shadow.ForeColor
    .OffsetX = shp.Shadow.OffsetX
    .OffsetY = shp.Shadow.OffsetY
    .Transparency = shp.Shadow.Transparency
    If shp.Shadow.Type <> -2 Then
        .Type = shp.Shadow.Type
    End If
End With
With shpNeu.WrapFormat
    .AllowOverlap = shp.WrapFormat.AllowOverlap
    .DistanceBottom = shp.WrapFormat.DistanceBottom
    .DistanceLeft = shp.WrapFormat.DistanceLeft
    .DistanceRight = shp.WrapFormat.DistanceRight
    .DistanceTop = shp.WrapFormat.DistanceTop
    .Side = shp.WrapFormat.Side
    .Type = shp.WrapFormat.Type
End With
If bDrehen Then shpNeu.Rotation = 180
End Sub

Function TrimZelle(s As String) As String
    s = Left(s, Len(s) - 2)
    TrimZelle = s
End Function

```



Die Beispieldatei *BspVIII\_02.dot* mit dem VBA-Code sowie das Seriendruck-Hauptdokument *BspVIII\_01.doc* und die WordArt-Vorlage *WordArt.doc* finden Sie auf der CD-ROM zum Buch im Ordner *\Beispiele\KapVIII*.



Abbildg. VIII.4 Die fertigen Tischkarten, bereit für den Ausdruck



## Zusammenfassung

Die Lösung in diesem Kapitel zeigt auf, wie mit dem Seriendruck und WordArt ein Satz dekorative Tischkarten erstellt werden können. Dabei werden aus dem Word-Objektmodell folgende Elemente veranschaulicht:

- Ansprechen des Seriendruckergebnisdokuments
- Das Festhalten und Festlegen von TextEffect- (WordArt-)Eigenschaften

